NAG Library Routine Document

F04EAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04EAF calculates the approximate solution of a set of real tridiagonal linear equations with a single right-hand side.

2 Specification

```
SUBROUTINE F04EAF(N, D, DU, DL, B, IFAIL)
INTEGER           N, IFAIL
double precision  D(N), DU(N), DL(N), B(N)
```

3 Description

F04EAF is based on the LINPACK routine SGTSL (see Dongarra *et al.* (1979)) and solves the equation

$$Tx = b,$$

where T is a real n by n tridiagonal matrix, by Gaussian elimination with partial pivoting. This routine should only be used if T is known not to be nearly singular since the routine only tests for exactly zero pivots.

4 References

Dongarra J J, Moler C B, Bunch J R and Stewart G W (1979) *LINPACK Users' Guide* SIAM, Philadelphia

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix T.

Constraint: N ≥ 1.

2: D(N) – ***double precision*** array *Input/Output*

On entry: the diagonal elements of T.

On exit: the diagonal elements of the upper triangular matrix, U, of the factorization of T.

3: DU(N) – ***double precision*** array *Input/Output*

On entry: the superdiagonal elements of T, stored in DU(2) to DU(n); DU(1) is not used.

On exit: the elements of the first superdiagonal of U, stored in DU(2) to DU(n).

4: DL(N) – ***double precision*** array *Input/Output*

On entry: the subdiagonal elements of T, stored in DL(2) to DL(n); DL(1) is not used.

On exit: the elements of the second superdiagonal of U, stored in DL(3) to DL(n).

5: B(N) – ***double precision*** array *Input/Output*

On entry: the right-hand side vector b.

On exit: B is overwritten by the solution vector x.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 1.

IFAIL > 1

The (IFAIL $- 1$)th diagonal element of U is zero. Unless IFAIL $=$ N $+ 1$, the factorization of T will not have been completed.

overflow

If overflow occurs during execution of F04EAF then either a diagonal element of U, while not being zero, is very small, or an element of b is very large. In the former case T is probably nearly singular and it would be advisable to use F01LEF, while in the latter case the equations should be scaled so that the elements of b are of a more moderate size.

underflow

Any underflows which occur during the execution of this routine are harmless, but if you who wish to avoid underflow and you may prefer to use the combination F01LEF and F04LEF to solve your tridiagonal equations.

7 Accuracy

The computed solution, say $\bar{x}$, of the equations $Tx = b$ will satisfy the equation

$$(T + E)\bar{x} = b,$$

where E can be expected to satisfy a bound of the form

$$\|E\| \leq \alpha\epsilon\|T\|,$$

α being a modest constant and ϵ being the ***machine precision***. The above result implies that the relative error in $\bar{x}$ satisfies the bound

$$\frac{\|\bar{x} - x\|}{\|\bar{x}\|} \leq c(T)\alpha\epsilon,$$

where $c(T)$ is the condition number of T with respect to inversion. Thus if T is nearly singular, x can be expected to have a high relative error.

8 Further Comments

The time taken by F04EAF is approximately proportional to n.

The tridiagonal equations $T^{\mathrm{T}}x = b$ may be solved by interchanging the parameters DU and DL in the call to this routine.

If you have multiple right-hand sides or are uncertain as to whether or not your matrix T is nearly singular, you are advised that the combination F01LEF and F04LEF can also be used to solve tridiagonal equations, but that this combination requires more storage and will usually be slower than F04EAF.

9 Example

This example solves the system of tridiagonal equations $Tx = b$, where

$$T = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} 2.7 \\ -0.5 \\ 2.6 \\ 0.6 \\ 2.7 \end{pmatrix}.$$

9.1 Program Text

```
*     F04EAF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NMAX
      PARAMETER        (NMAX=100)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, N
*     .. Local Arrays ..
      DOUBLE PRECISION B(NMAX), D(NMAX), DL(NMAX), DU(NMAX)
*     .. External Subroutines ..
      EXTERNAL         F04EAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F04EAF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.LT.1 .OR. N.GT.NMAX) THEN
         WRITE (NOUT,99999) 'N is out of range: N = ', N
      ELSE
         READ (NIN,*) (D(I),I=1,N)
         IF (N.GT.1) THEN
            READ (NIN,*) (DU(I),I=2,N)
            READ (NIN,*) (DL(I),I=2,N)
         END IF
         READ (NIN,*) (B(I),I=1,N)
         IFAIL = 1
*
         CALL F04EAF(N,D,DU,DL,B,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,*) 'Solution vector'
            WRITE (NOUT,99998) (B(I),I=1,N)
         ELSE
            WRITE (NOUT,99997) IFAIL
         END IF
      END IF
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,5F9.3)
99997 FORMAT (1X,' ** F04EAF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
F04EAF Example Program Data
   5
  3.0   2.3  -5.0  -0.9   7.1
  2.1  -1.0   1.9   8.0
  3.4   3.6   7.0  -6.0
  2.7  -0.5   2.6   0.6   2.7
```

9.3 Program Results

```
 F04EAF Example Program Results

 Solution vector
    -4.000    7.000    3.000   -4.000   -3.000
```